

9/6.3

Fonction HELP

La fonction **HELP** peut rendre de grands services aux joueurs de jeux d'aventures : elle donne la liste des mots clés « compris » par la machine.

Pour accéder à cette liste, il suffit d'appuyer simultanément sur les touches **Ctrl** et **H**. Les touches flèches vers le haut et vers le bas font défiler les mots clés dans une fenêtre de texte située en bas de l'écran.

La fenêtre texte est définie grâce à la macro **TXT WIN ENABLE** (Cf. point d'entrée **OBB66H** du Firmware) puis effacée grâce à la macro **TXT CLEAR WIN** (Cf. point d'entrée **OBB6CH** du Firmware) [voir Partie 4, Chap. 2.7, pages 13 et 14].

Les trois premiers mots clés sont affichés grâce aux macros **TXT SET-CURSOR** et **TXT OUTPUT** (Cf. points d'entrées **OBB75H** et **OBB5AH** du Firmware) [Voir Partie 4, Chap. 2.7 pages 12 et 14].


Le programme se met alors en attente d'une action au clavier :

- si la séquence **Ctrl H** est reconnue, la fenêtre **HELP** est effacée, et le contrôle est redonné au programme principal ;
- si la touche flèche vers le haut est reconnue, et s'il reste des mots clés à lister, un scrolling vers le haut est exécuté ;
- si la touche flèche vers le bas est reconnue, et si le premier mot clé ne se trouve pas sur la première ligne de la fenêtre, un scrolling vers le bas est exécuté.

Les autres touches du clavier n'ont aucun effet.

Le scrolling est obtenu grâce à la macro **ROLL**. Pour plus de détails, reportez-vous au point d'entrée **OBC50H** du Firmware (voir Partie 4, Chap. 2.7 page 38).

Les actions effectuées par le programme correspondent à la logique de l'ordinogramme suivant :



Le listing du programme Assembleur est le suivant :

```

1          ;Fonction HELP
2          ;I/O: Aucune,
3          ;Pt d'entree: HELP
4          ;-----
5          ORG  920CH
6          LOAD 920CH
7          ADCOUR:  DS  2
8          L1:      DS  1
9          SS:      DS  2
10         CUROFF:  EQU  OBB84H          ;TXT CUR OFF
11         SETCUR:  EQU  OBB75H          ;TXT SET CURSOR
12         TXTOUT:  EQU  OBB5AH          ;TXT OUTPUT
13         WAITCH:  EQU  OBB06H          ;TXT WAIT CHAR
14         DEFWIN:  EQU  OBB66H          ;Def. Fenetre
15         EFFWIN:  EQU  OBB6CH          ;Eff. Fenetre
16         ROLL:    EQU  OBC50H          ;SCR SW ROLL
17         ;-----
18         HELP:    EQU  $
19 9211 211600      LD  HL,16H
20 9214 111827      LD  DE,2718H
21 9217 CD66BB      CALL DEFWIN          ;Definition Fenetre
22 921A CD6CBB      CALL EFFWIN          ;Effacement Fenetre
23 921D CDB4BB      CALL CUROFF
24 9220 CD6CBB      CALL EFFWIN          ;Effacement Fenetre
25 9223 210080      LD  HL,8000H
26 9226 220C92      LD  (ADCOUR),HL      ;à debut verbes
27 9229 211701      LD  HL,117H
28 922C CD75BB      CALL SETCUR
29 922F 2A0C92      LD  HL,(ADCOUR)
30 9232 CDFE92      CALL AF251          ;Affichage liane 1

```

Partie 9 : Programmes

```

31 9235 2A0C92      LD   HL, (ADCOUR)
32 9238 23          INC  HL
33 9239 220C92      LD   (ADCOUR),HL
34 923C 220F92      LD   (SS),HL
35 923F 211801      LD   HL,118H
36 9242 CD75BB      CALL SETCUR
37 9245 2A0C92      LD   HL, (ADCOUR)
38 9248 CDFE92      CALL AF251          ;Affichage ligne 2
39 924B 2A0C92      LD   HL, (ADCOUR)
40 924E 23          INC  HL
41 924F 220C92      LD   (ADCOUR),HL
42 9252 211901      LD   HL,119H
43 9255 CD75BB      CALL SETCUR
44 9258 2A0C92      LD   HL, (ADCOUR)
45 925B CDFE92      CALL AF251          ;Affichage ligne 3
46                ;
47 925E 3E00        LD   A,0
48 9260 320E92      LD   (L1),A
49 9263 2600        LD   H,0
50 9265 1627        LD   D,39
51 9267 2E16        LD   L,22
52 9269 1E19        LD   E,25
53 926B D5          PUSH DE
54 926C E5          PUSH HL
55 926D 2A0F92      LD   HL, (SS)
56 9270 220C92      LD   (ADCOUR),HL
57                AA1: EQU  $
58 9273 CD06BB      CALL WAITCH        ;Attente clavier
59 9276 FE08        CP   8
60 927B 2B65        JR   Z,AA5        ;CTRL H (Fin du HELP)

```

```

61 927A FEF0          CP   OF0H
62 927C 2806          JR   Z,AA2          ; ->
63 927E FEF1          CP   OF1H
64 9280 282D          JR   Z,AA3          ; <-
65 9282 18EF          JR   AA1          ;What ?
66                   ;
67                   AA2: EQU $          ;Gestion touche UP
68 9284 3A0E92        LD   A,(L1)
69 9287 FE0A          CP   10          ;Nombre max de mots
70 9289 28E8          JR   Z,AA1
71 928B 3C           INC  A
72 928C 320E92        LD   (L1),A
73 928F 0601          LD   B,1
74 9291 E1           POP  HL
75 9292 D1           POP  DE
76 9293 3E00          LD   A,0
77 9295 D5           PUSH DE
78 9296 E5           PUSH HL
79 9297 CD30BC        CALL ROLL
80 929A CDE592        CALL VESU
81 929D CDE592        CALL VESU
82 92A0 CDF592        CALL AFVE25
83 92A3 CD1C93        CALL VEPR
84 92A6 CD1C93        CALL VEPR
85 92A9 23           INC  HL
86 92AA 220C92        LD   (ADCOUR),HL
87 92AD 18C4          JR   AA1
88                   ;
89                   AA3: EQU $          ;Gestion touche DOWN
90 92AF 3A0E92        LD   A,(L1)

```

```

91 92B2 FE00          CP    0
92 92B4 28BD          JR    Z,AA1
93 92B6 3D            DEC   A
94 92B7 320E92        LD    (L1),A
95 92BA 0600          LD    B,0
96 92BC E1            POP   HL
97 92BD D1            POP   DE
98 92BE 3E00          LD    A,0
99 92C0 D5            PUSH  DE
100 92C1 E5           PUSH  HL
101 92C2 CD50BC        CALL  ROLL
102 92C5 2A0C92        LD    HL,(ADCOUR)
103 92C8 2B            DEC   HL
104 92C9 220C92        LD    (ADCOUR),HL
105 92CC CD1C93        CALL  VEPR
106 92CF CD1C93        CALL  VEPR
107 92D2 23            INC   HL
108 92D3 220C92        LD    (ADCOUR),HL
109 92D6 CD1193        CALL  AFVE23
110 92D9 23            INC   HL
111 92DA 220C92        LD    (ADCOUR),HL
112 92DD 1894          JR    AA1
113                   ;
114                   AA5:   EQU   $           ;Gestion touche CTRL+H
115 92DF D1            POP   DE
116 92E0 E1            POP   HL
117 92E1 CD6CBB        CALL  EFFWIN
118 92E4 C9            RET           ;Retour au prog appelant
119                   ;
120                   VESU:  EQU   $           ;Verbe suivant

```

```

121 92E5 2A0C92          LD   HL, (ADCOUR)
122                      VESU1: EQU  $
123 92E8 7E             LD   A, (HL)
124 92E9 FEFF          CP   OFFH
125 92EB 2803          JR   Z, VESU2
126 92ED 23            INC  HL
127 92EE 18F8          JR   VESU1
128                      VESU2: EQU  $
129 92F0 23            INC  HL
130 92F1 220C92        LD   (ADCOUR), HL
131 92F4 09            RET
132                      ;
133                      AFVE25: EQU  $                ;Affichage ligne 25
134 92F5 210301        LD   HL, 103H
135 92F8 CD75BB        CALL SETCUR
136 92FB 2A0C92        LD   HL, (ADCOUR)
137                      AF251: EQU  $
138 92FE 7E             LD   A, (HL)
139 92FF FEFF          CP   OFFH
140 9301 280A          JR   Z, AF252
141 9303 FE20          CP   32                ;Si <32 (blanc)
142 9305 3803          JR   C, AFSU           ;Caract. non affiche
143 9307 CD5ABB        CALL TXTOUT           ;Aff. d'un caractere
144                      AFSU: EQU  $
145 930A 23            INC  HL
146 930B 18F1          JR   AF251
147                      AF252: EQU  $
148 930D 220C92        LD   (ADCOUR), HL
149 9310 09            RET
150                      ;

```

```

151          AFVE23:    EQU  $                ;Affichage ligne 23
152 9311 210101      LD   HL,101H
153 9314 CD75BB      CALL SETCUR
154 9317 2A0C92      LD   HL,(ADCOUR)
155 931A 18E2       JR   AF251

156          ;
157          VEPR:     EQU  $                ;Verbe precedent
158 931C 2A0C92      LD   HL,(ADCOUR)
159 931F 2B         DEC  HL
160          VEPR1:   EQU  $
161 9320 7E         LD   A,(HL)
162 9321 FEFF       CP   OFFH
163 9323 2803       JR   Z,VEPR2
164 9325 2B         DEC  HL
165 9326 18F8       JR   VEPR1
166          VEPR2:   EQU  $
167 9328 220C92      LD   (ADCOUR),HL
168 932B C9         RET
169                END

```

- Lignes 19 à 21 : Définition de la fenêtre.
- Ligne 24 : Effacement de la fenêtre.
- Lignes 25 à 56 : Affichage des trois premières lignes.
- Ligne 58 : Attente d'une action au clavier.
- Ligne 60 : Ctrl H.
- Ligne 62 : Up.
- Ligne 64 : Down.
- Lignes 67 à 87 : Gestion du Up.
- Lignes 89 à 112 : Gestion du Down.
- Lignes 114 à 118 : Gestion du Ctrl H.
- Lignes 120 à 131 : Modification du pointeur de mot clé DOWN.
- Lignes 134 à 149 : Affichage en ligne 25.
- Lignes 151 à 155 : Affichage en ligne 23.
- Lignes 157 à 168 : Modification du pointeur de mot clé UP.

Les données affichées par le **HELP** vont être implantées par défaut en &H8000. Cette adresse peut être modifiée ligne 25. Le premier octet

(d'adresse &H8000 dans notre cas) doit contenir la valeur zéro. Les mots à afficher doivent être séparés par le caractère séparateur &HFF :

```
00 Do1 FF Do2 FF ... DoN FF
```

où Do1 à DoN représentent les mots clés à afficher.

Enfin, le nombre de scrolling (c'est-à-dire le nombre de données moins trois) doit être initialisé ligne 69. Il est égal à 10 dans notre exemple.

Il est également possible de modifier l'adresse d'implantation et le nombre de scrolling sans modifier le code source du listing Assembleur en utilisant trois instructions Basic Poke :

```
POKE &H9224,LSB
POKE &H9225,MSB
POKE &H9288,SC
```

où LSB est le poids faible de l'adresse d'implantation des mots clés,
MSB est le poids fort de l'adresse d'implantation des mots clés,
SC est le nombre de scrolling. Ce nombre est égal au nombre de mots clés moins trois : par exemple 4 pour 7 mots clés.

Le programme Assembleur qui vient d'être décrit peut également être exécuté en utilisant le chargeur Basic suivant :

```
1000 FOR I=&9211 TO &932B
1010 READ A#
1020 A=VAL("&"+A#)
1030 POKE I,A
1040 NEXT I
1060 -----
1070 DATA 21,16,0,11,18,27,CD,66,BB,CD,6C,BB,CD,84,BB,CD
1080 DATA 6C,BB,21,0,80,22,C,92,21,17,1,CD,75,BB,2A,C
1090 DATA 92,CD,FE,92,2A,C,92,23,22,C,92,22,F,92,21,18
1100 DATA 1,CD,75,BB,2A,C,92,CD,FE,92,2A,C,92,23,22,C
1110 DATA 92,21,19,1,CD,75,BB,2A,C,92,CD,FE,92,3E,0,32
1120 DATA E,92,26,0,16,27,2E,16,1E,19,D5,E5,2A,F,92,22
1130 DATA C,92,CD,6,BB,FE,8,28,65,FE,FO,28,6,FE,F1,28
1140 DATA 2D,18,EF,3A,E,92,FE,A,28,E8,3C,32,E,92,6,1
1150 DATA E1,D1,3E,0,D5,E5,CD,50,BC,CD,E5,92,CD,E5,92,CD
1160 DATA F5,92,CD,1C,93,CD,1C,93,23,22,C,92,18,C4,3A,E
1170 DATA 92,FE,0,28,BD,3D,32,E,92,6,0,E1,D1,3E,0,D5
1180 DATA E5,CD,50,BC,2A,C,92,2B,22,C,92,CD,1C,93,CD,1C
1190 DATA 93,23,22,C,92,CD,11,93,23,22,C,92,18,94,D1,E1
1200 DATA CD,6C,BB,C9,2A,C,92,7E,FE,FF,28,3,23,18,F8,23
1210 DATA 22,C,92,C9,21,3,1,CD,75,BB,2A,C,92,7E,FE,FF
1220 DATA 28,A,FE,20,38,3,CD,5A,BB,23,18,F1,22,C,92,C9
1230 DATA 21,1,1,CD,75,BB,2A,C,92,18,E2,2A,C,92,2B,7E
1240 DATA FE,FF,28,3,2B,18,F8,22,C,92,C9,0,0,0,0,0
```

Les données de checksum de ce programme sont les suivantes :

```
49 F8 9B 42 65 29 F9 40 E2 8C 55 DC 2E 88 F4 28 58 F0
```

Reportez-vous à la Partie 9, chapitre 8.4, pour vérifier la cohérence des données entrées dans les lignes de DATA du chargeur.

